# Maintainer: Angel Velasquez <angvp@archlinux.org>  
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgbase=python-cairo
pkgname=(python-cairo python2-cairo)
pkgver=1.17.0
pkgrel=2
pkgdesc="Python bindings for the cairo graphics library"
arch=('x86_64')
url="https://pycairo.readthedocs.io/en/latest/"
license=('LGPL2.1' 'MPL')
makedepends=(python python2 cairo git)
_commit=910ee82a5cd6fc75c23a893826d5976d8e862ea1  # tags/v1.17.0
source=("git+https://github.com/pygobject/pycairo/#commit=$_commit")
sha256sums=('SKIP')

pkgver() {
  cd pycairo
  git describe --tags | sed 's/^v//;s/-/+/g'
}


build() {
  cd pycairo
  python2 setup.py build
  python setup.py build
}

package_python2-cairo() {
  depends=('cairo' 'python2')
  pkgdesc="Python2 bindings for the cairo graphics library"

  cd pycairo
  python2 setup.py install --skip-build --root="${pkgdir}" --optimize='1'
}

package_python-cairo() {
  depends=('cairo' 'python')

  cd pycairo
  python setup.py install --skip-build --root="${pkgdir}" --optimize='1'
}
